/**
 * @(#) FiscalYearsProxy.as
 *
 * @date		2010. 11. 01
 * @version		2.0
 * @author		Shin, Yunwoo
 * @client		Service Ace 
 * @module		Compensation  (급여)
 * @description	년도별 급여조회를 위한 회계년도 조회- Proxy
 * 				
 */
package com.skt.common_hr.compensation.model
{
	import com.skt.common_hr.common.Variables;
	import com.skt.common_hr.common.puremvc.CommonProxy;
	import com.skt.common_hr.compensation.CompensationFacade;
	
	import mx.collections.ArrayCollection;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;
	
	public class FiscalYearsProxy extends CommonProxy
	{
		public static const NAME:String = "FiscalYearsProxy";
		public var yearList:ArrayCollection = new ArrayCollection()
		
		/**
		 * 생성자
		 * 
		 * @param
		 * @return void
		 */
		public function FiscalYearsProxy(data:Object=null)
		{
			super(NAME, data);
		}
		
		/**
		 * data load를 수행
		 * 		- DataLoader 를 생성하고 Import Parameter 객체를 생성, RFC를 호출
		 * 
		 * @param pernr 사번
		 * @return void
		 */
		public function load(pernr:String):void
		{
			var param:Object = new Object();
			param["I_PERNR"] = pernr;					// 사번
			
			loadData("Z_HR_0005", param, loadResult, loadFault);
		}
		



		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Event Listeners
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 정상적으로 리턴된 결과를 처리
		 * 
		 * @param
		 * @return void
		 */
		public function loadResult(event:ResultEvent):void
		{
			if(retCode==0) {  
				yearList = parseYears(event.result.T_RESULT);
				facade.sendNotification(CompensationFacade.LOAD_FISCALYEARS_COMPLETE, this);
			}
			else {
				facade.sendNotification(CompensationFacade.LOAD_FISCALYEARS_FAILED, this);
			}
		}
		
		/**
		 * 서버에서 오류가 발생했을 처리
		 * 
		 * @param
		 * @return void
		 */
		private function loadFault(event:FaultEvent):void
		{
			facade.sendNotification(CompensationFacade.LOAD_FISCALYEARS_FAILED, this);
		}


		///////////////////////////////////////////////////////////////////////////////////////////////////
		//
		//	Data Parsing functions
		//
		///////////////////////////////////////////////////////////////////////////////////////////////////
		/**
		 * 서버에서 전달된 result 객체를 분석하여 Value Object를 생성
		 * 
		 * @param
		 * @return 인사기록 Value Object
		 */
		private function parseYears(srvTable:ArrayCollection):ArrayCollection
		{
			var coll:ArrayCollection = new ArrayCollection();
			for each(var rec:Object in srvTable)
			{
				coll.addItem(rec.PYEAR.toString());
			}
			return coll;
		}
	}
}